PRE-FIT

PURPOSE

Determines the “best fit” coefficients for a model, where the search is restricted to the points of a user-defined lattice.

DESCRIPTION

The desired lattice is defined by the FOR statements at the end of the command. The “best fit” coefficients are determined by systematically evaluating the sum of squares surface at each of the points of the specified lattice of parameter values. The minimum point over the lattice is determined and noted, along with other extrema information. This command receives only light usage, typically in the following 2 contexts:

1. immediately before the FIT command when the model is nonlinear, when the starting values are unknown, and when the analyst is interested in gathering some preliminary information about the nature of the sum of squares surface. Although the DATAPLOT nonlinear FIT command usually converges even without specifying starting values, there are cases where it does not. The PRE-FIT can help determine adequate starting values.
2. immediately after the FIT command when a least squares fit has been carried out, and the analyst is interested in examining the nature of the Lp surface (for p not = 2) in the vicinity of the least squares estimate. In this context, the FIT POWER command is also used. This usage is discussed in more detail in the documentation for the FIT POWER command.

SYNTAX

PRE-FIT <y1> = <f> FOR <p> = <p1> <pinc> <p2> FOR <q> = <q1> <qinc> <q2>
<additional optional FOR statements>

where <y1> is the response (= dependent) variable;
<f> is

1. any general FORTRAN-like expression; or
2. any function name that the user has already created via the LET FUNCTION command);

<p> is the name of a parameter in the function;
<p1> is a number or parameter that specifies the minimum value for the parameter;
<pinc> is a number or parameter that specifies the increment for the parameter;
<p2> is a number or parameter that specifies the maximum value for the parameter;
<q> is the name of another parameter in the function;
<q1> is a number or parameter that specifies the minimum value for the parameter;
<qinc> is a number or parameter that specifies the increment for the parameter;
<q2> is a number or parameter that specifies the maximum value for the parameter;
and any additional FOR statements have a similar syntax.

All of the FOR statements should be on the same line.

EXAMPLES

```
PRE-FIT Y=A+B*EXP(-C*X) FOR A=10 1 20 FOR B=50 10 100 FOR C = .1 .1 1
PRE-FIT Y = A+EXP(C/X) FOR A = 1 2 10 FOR C = .5 .05 .6
```

DEFAULT

None

SYNONYMS

None

RELATED COMMANDS

PRED	=	A variable where predicted values are stored.
RES	=	A variable where residuals are stored.
RESSD	=	A parameter where the residual standard deviation is stored.
RESDF	=	A parameter where the residual degrees of freedom is stored.
REPSD	=	A parameter where the replication standard deviation is stored.
REPDF	=	A parameter where the replication degrees of freedom is stored.
LOFCDF	=	A parameter where the lack of fit cdf is stored.
FIT	=	Carries out a least squares fit.
EXACT RATIONAL FIT	=	Carries out an exact rational fit.

SPLINE FIT	=	Carries out a spline fit.
SMOOTH	=	Carries out a smoothing.
ANOVA	=	Carries out an ANOVA.
MEDIAN POLISH	=	Carries out a median polish.
PLOT	=	Generates a data or function plot.

APPLICATIONS

Fitting

IMPLEMENTATION DATE

Pre-1987

PROGRAM

```
. DAN CHWIRUT ULTRASONIC REFERENCE BLOCK ANALYSIS
. PERFORM A PRE-FIT AND THEN A NON-LINEAR REGRESSION
SKIP 25
READ CHWIRUT1.DAT Y X
LET FUNCTION F = A + B*EXP(-C*X)
PRE-FIT Y=F FOR A = 0 10 50 FOR B = 100 20 200 FOR C = 0.5 0.5 2
FIT Y = F
.
MULTIPLOT 2 2
MULTIPLOT CORNER COORDINATES 0 0 100 100
TITLE ORIGINAL DATA
CHARACTERS X ALL
X1LABEL METAL DISTANCE
Y1LABEL ULTRASONIC RESPONSE
PLOT Y X X
TITLE PREDICTED VALUES
LINE BLANK SOLID
CHARACTER X BLANK
PLOT Y PRED VS X
TITLE RESIDUALS
Y1LABEL
PLOT RES VS X
X1LABEL
TITLE NORMAL PROBABILITY PLOT
NORMAL PROBABILITY PLOT RES
END OF MULTIPLOT
```

The following output is generated.

```
 INPUT FUNCTION         = A+B*EXP(-C*X)
 OUTPUT FUNCTION        = A+B*EXP(-C*X)

 THE NAME F         HAS JUST BEEN EQUIVALENCED
 TO THE FUNCTION       -- A+B*EXP(-C*X)
```

```
LEAST SQUARES NON-LINEAR PRE-FIT
      SAMPLE SIZE N =      214
      MODEL--Y=(A + B*EXP(-C*X))
      REPLICATION CASE
      REPLICATION STANDARD DEVIATION =     0.3281762600D+01
      REPLICATION DEGREES OF FREEDOM =         192
      NUMBER OF DISTINCT SUBSETS     =          22

      LATTICE VALUES FOR A        =   0.0000000E+00  0.1000000E+02  0.5000000E+02
      LATTICE VALUES FOR B        =   0.1000000E+03  0.2000000E+02  0.2000000E+03
      LATTICE VALUES FOR C        =   0.5000000E+00  0.5000000E+00  0.2000000E+01

      NUMBER OF LATTICE POINTS    =         144

  STEP                  RESIDUAL  *  PARAMETER
 NUMBER                 STANDARD  *  ESTIMATES
                        DEVIATION *
---------------------------------*-----------
    1--               0.89193E+01 * 0.00000E+00 0.10000E+03 0.50000E+00
    6--               0.81236E+01 * 0.00000E+00 0.12000E+03 0.10000E+01
   10--               0.67238E+01 * 0.00000E+00 0.14000E+03 0.10000E+01
   26--               0.49619E+01 * 0.10000E+02 0.10000E+03 0.10000E+01
   30--               0.44007E+01 * 0.10000E+02 0.12000E+03 0.10000E+01

       FINAL PARAMETER ESTIMATES
       1  A                   10.0000
       2  B                   120.000
       3  C                   1.00000

      RESIDUAL    STANDARD DEVIATION =         4.4007382393
      RESIDUAL    DEGREES OF FREEDOM =         211
      REPLICATION STANDARD DEVIATION =         3.2817625999
      REPLICATION DEGREES OF FREEDOM =         192
      LACK OF FIT F RATIO =  1327.1779 = THE 100.0000% POINT OF THE
      F DISTRIBUTION WITH     19 AND    192 DEGREES OF FREEDOM

LEAST SQUARES NON-LINEAR FIT
      SAMPLE SIZE N =      214
      MODEL--Y =(A + B*EXP(-C*X))
      REPLICATION CASE
      REPLICATION STANDARD DEVIATION =     0.3281762600D+01
      REPLICATION DEGREES OF FREEDOM =         192
      NUMBER OF DISTINCT SUBSETS     =          22

ITERATION  CONVERGENCE  RESIDUAL  *  PARAMETER
 NUMBER      MEASURE    STANDARD  *  ESTIMATES
                        DEVIATION *
---------------------------------*-----------
    1--  0.10000E-01  0.44007E+01 * 0.10000E+02 0.12000E+03 0.10000E+01
    2--  0.50000E-02  0.33695E+01 * 0.70832E+01 0.11387E+03 0.94537E+00
    3--  0.25000E-02  0.33691E+01 * 0.69854E+01 0.11389E+03 0.94192E+00

       FINAL PARAMETER ESTIMATES           (APPROX. ST. DEV.)    T VALUE
       1  A                   6.98114       (0.5187    )          13.
       2  B                   113.877       ( 1.753    )          65.
       3  C                  0.941724       (0.2553E-01)          37.
```

```
RESIDUAL    STANDARD DEVIATION =         3.3691074848
RESIDUAL    DEGREES OF FREEDOM =         211
REPLICATION STANDARD DEVIATION =         3.2817625999
REPLICATION DEGREES OF FREEDOM =         192
LACK OF FIT F RATIO =       1.5990 = THE  94.0437% POINT OF THE
F DISTRIBUTION WITH     19 AND    192 DEGREES OF FREEDOM



COEF AND SD(COEF) WRITTEN TO FILE DPST1F.DAT
```